unittest
{
	assert( [ "4 4 1", "2 1", "1 3", "3 2", "3 4", "4 1" ].solve == "0 1 0 1" );
	assert( [ "5 10 0", "1 2", "1 3", "1 4", "1 5", "3 2", "2 4", "2 5", "4 3", "5 3", "4 5" ].solve == "0 0 0 0 0" );
	assert( [ "10 9 3", "10 1", "6 7", "8 2", "2 5", "8 4", "7 3", "10 9", "6 4", "5 8", "2 6", "7 5", "3 1" ].solve == "1 3 5 4 3 3 3 3 1 0" );
}



import std.conv;
import std.format;
import std.range;
import std.stdio;

void main()
{
	stdin.byLineCopy.solve.writeln;
}

auto solve( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = 0L;
	auto m = 0L;
	auto k = 0L;
	input.front.formattedRead!"%s %s %s"( n, m, k );
	auto abs = new long[ 2 ][ m ];
	foreach( ref ab; abs )
	{
		input.popFront;
		ab = input.front.split.to!( long[] );
		ab[] -= 1;
	}
	auto cdss = new long[][ n ];
	foreach( i; 0L .. k )
	{
		input.popFront;
		auto cd = input.front.split.to!( long[] );
		cd[] -= 1;
		cdss[ cd[ 0 ] ] ~= cd[ 1 ];
		cdss[ cd[ 1 ] ] ~= cd[ 0 ];
	}
	
	// 友達ネットワークを構築、友達の数をカウント
	auto uft = UnionFindTree( n );
	auto fs = new long[ n ];
	foreach( ab; abs )
	{
		uft.unite( ab[ 0 ], ab[ 1 ] );
		fs[ ab[ 0 ] ]++;
		fs[ ab[ 1 ] ]++;
	}
	
	// 友達候補の数を計算
	auto anss = new long[ n ];
	foreach( i, ref ans, cds; lockstep( anss, cdss ) )
	{
		// 友達ネットワークから自分と友達を除いたものが友達候補
		ans = uft.weight( i ) - 1 - fs[ i ];
		
		// 友達候補からブロック対象も除外
		foreach( cd; cds )
		{
			if( uft.same( i, cd ) ) ans--;
		}
	}
	return anss.to!( string[] ).join( " " );
}

// UnionFind木
struct UnionFindTree
{
	long[] ps;
	long[] rs;
	long   rn;
	
	this( long n )
	{
		ps.length = n;
		ps[] = -1;
		rs.length = n;
		rn = n;
	}
	
	long root( long i )
	{
		if( ps[ i ] < 0 ) return i;
		else              return ps[ i ] = root( ps[ i ] );
	}
	
	auto same( long i, long j )
	{
		return root( i ) == root( j );
	}
	
	auto unite( long i, long j )
	{
		i = root( i );
		j = root( j );
		if( i == j ) return;
		if( rs[ i ] < rs[ j ] )
		{
			ps[ j ] += ps[ i ];
			ps[ i ] = j;
		}
		else
		{
			ps[ i ] += ps[ j ];
			ps[ j ] = i;
			if( rs[ i ] == rs[ j ] ) rs[ i ]++;
		}
		rn--;
	}
	
	auto weight( long i )
	{
		return -ps[ root( i ) ];
	}
	
	auto toString()
	{
		return ps.to!( string );
	}
}
